// SPDX-License-Identifier: MIT

pragma solidity 0.8.19;

// OpenZeppelin Contracts (last updated v4.6.0) (token/ERC20/IERC20.sol)

/**
 * @dev Interface of the ERC20 standard as defined in the EIP.
 */
interface IERC20 {
    /**
     * @dev Emitted when `value` tokens are moved from one account (`from`) to
     * another (`to`).
     *
     * Note that `value` may be zero.
     */
    event Transfer(address indexed from, address indexed to, uint256 value);

    /**
     * @dev Emitted when the allowance of a `spender` for an `owner` is set by
     * a call to {approve}. `value` is the new allowance.
     */
    event Approval(address indexed owner, address indexed spender, uint256 value);

    /**
     * @dev Returns the amount of tokens in existence.
     */
    function totalSupply() external view returns (uint256);

    /**
     * @dev Returns the amount of tokens owned by `account`.
     */
    function balanceOf(address account) external view returns (uint256);

    /**
     * @dev Moves `amount` tokens from the caller's account to `to`.
     *
     * Returns a boolean value indicating whether the operation succeeded.
     *
     * Emits a {Transfer} event.
     */
    function transfer(address to, uint256 amount) external returns (bool);

    /**
     * @dev Returns the remaining number of tokens that `spender` will be
     * allowed to spend on behalf of `owner` through {transferFrom}. This is
     * zero by default.
     *
     * This value changes when {approve} or {transferFrom} are called.
     */
    function allowance(address owner, address spender) external view returns (uint256);

    /**
     * @dev Sets `amount` as the allowance of `spender` over the caller's tokens.
     *
     * Returns a boolean value indicating whether the operation succeeded.
     *
     * IMPORTANT: Beware that changing an allowance with this method brings the risk
     * that someone may use both the old and the new allowance by unfortunate
     * transaction ordering. One possible solution to mitigate this race
     * condition is to first reduce the spender's allowance to 0 and set the
     * desired value afterwards:
     * https://github.com/ethereum/EIPs/issues/20#issuecomment-263524729
     *
     * Emits an {Approval} event.
     */
    function approve(address spender, uint256 amount) external returns (bool);

    /**
     * @dev Moves `amount` tokens from `from` to `to` using the
     * allowance mechanism. `amount` is then deducted from the caller's
     * allowance.
     *
     * Returns a boolean value indicating whether the operation succeeded.
     *
     * Emits a {Transfer} event.
     */
    function transferFrom(
        address from,
        address to,
        uint256 amount
    ) external returns (bool);
}

// OpenZeppelin Contracts (last updated v4.8.0) (token/ERC20/utils/SafeERC20.sol)

// OpenZeppelin Contracts v4.4.1 (token/ERC20/extensions/draft-IERC20Permit.sol)

/**
 * @dev Interface of the ERC20 Permit extension allowing approvals to be made via signatures, as defined in
 * https://eips.ethereum.org/EIPS/eip-2612[EIP-2612].
 *
 * Adds the {permit} method, which can be used to change an account's ERC20 allowance (see {IERC20-allowance}) by
 * presenting a message signed by the account. By not relying on {IERC20-approve}, the token holder account doesn't
 * need to send a transaction, and thus is not required to hold Ether at all.
 */
interface IERC20Permit {
    /**
     * @dev Sets `value` as the allowance of `spender` over ``owner``'s tokens,
     * given ``owner``'s signed approval.
     *
     * IMPORTANT: The same issues {IERC20-approve} has related to transaction
     * ordering also apply here.
     *
     * Emits an {Approval} event.
     *
     * Requirements:
     *
     * - `spender` cannot be the zero address.
     * - `deadline` must be a timestamp in the future.
     * - `v`, `r` and `s` must be a valid `secp256k1` signature from `owner`
     * over the EIP712-formatted function arguments.
     * - the signature must use ``owner``'s current nonce (see {nonces}).
     *
     * For more information on the signature format, see the
     * https://eips.ethereum.org/EIPS/eip-2612#specification[relevant EIP
     * section].
     */
    function permit(
        address owner,
        address spender,
        uint256 value,
        uint256 deadline,
        uint8 v,
        bytes32 r,
        bytes32 s
    ) external;

    /**
     * @dev Returns the current nonce for `owner`. This value must be
     * included whenever a signature is generated for {permit}.
     *
     * Every successful call to {permit} increases ``owner``'s nonce by one. This
     * prevents a signature from being used multiple times.
     */
    function nonces(address owner) external view returns (uint256);

    /**
     * @dev Returns the domain separator used in the encoding of the signature for {permit}, as defined by {EIP712}.
     */
    // solhint-disable-next-line func-name-mixedcase
    function DOMAIN_SEPARATOR() external view returns (bytes32);
}

// OpenZeppelin Contracts (last updated v4.8.0) (utils/Address.sol)

/**
 * @dev Collection of functions related to the address type
 */
library Address {
    /**
     * @dev Returns true if `account` is a contract.
     *
     * [IMPORTANT]
     * ====
     * It is unsafe to assume that an address for which this function returns
     * false is an externally-owned account (EOA) and not a contract.
     *
     * Among others, `isContract` will return false for the following
     * types of addresses:
     *
     *  - an externally-owned account
     *  - a contract in construction
     *  - an address where a contract will be created
     *  - an address where a contract lived, but was destroyed
     * ====
     *
     * [IMPORTANT]
     * ====
     * You shouldn't rely on `isContract` to protect against flash loan attacks!
     *
     * Preventing calls from contracts is highly discouraged. It breaks composability, breaks support for smart wallets
     * like Gnosis Safe, and does not provide security since it can be circumvented by calling from a contract
     * constructor.
     * ====
     */
    function isContract(address account) internal view returns (bool) {
        // This method relies on extcodesize/address.code.length, which returns 0
        // for contracts in construction, since the code is only stored at the end
        // of the constructor execution.

        return account.code.length > 0;
    }

    /**
     * @dev Replacement for Solidity's `transfer`: sends `amount` wei to
     * `recipient`, forwarding all available gas and reverting on errors.
     *
     * https://eips.ethereum.org/EIPS/eip-1884[EIP1884] increases the gas cost
     * of certain opcodes, possibly making contracts go over the 2300 gas limit
     * imposed by `transfer`, making them unable to receive funds via
     * `transfer`. {sendValue} removes this limitation.
     *
     * https://diligence.consensys.net/posts/2019/09/stop-using-soliditys-transfer-now/[Learn more].
     *
     * IMPORTANT: because control is transferred to `recipient`, care must be
     * taken to not create reentrancy vulnerabilities. Consider using
     * {ReentrancyGuard} or the
     * https://solidity.readthedocs.io/en/v0.5.11/security-considerations.html#use-the-checks-effects-interactions-pattern[checks-effects-interactions pattern].
     */
    function sendValue(address payable recipient, uint256 amount) internal {
        require(address(this).balance >= amount, "Address: insufficient balance");

        (bool success, ) = recipient.call{value: amount}("");
        require(success, "Address: unable to send value, recipient may have reverted");
    }

    /**
     * @dev Performs a Solidity function call using a low level `call`. A
     * plain `call` is an unsafe replacement for a function call: use this
     * function instead.
     *
     * If `target` reverts with a revert reason, it is bubbled up by this
     * function (like regular Solidity function calls).
     *
     * Returns the raw returned data. To convert to the expected return value,
     * use https://solidity.readthedocs.io/en/latest/units-and-global-variables.html?highlight=abi.decode#abi-encoding-and-decoding-functions[`abi.decode`].
     *
     * Requirements:
     *
     * - `target` must be a contract.
     * - calling `target` with `data` must not revert.
     *
     * _Available since v3.1._
     */
    function functionCall(address target, bytes memory data) internal returns (bytes memory) {
        return functionCallWithValue(target, data, 0, "Address: low-level call failed");
    }

    /**
     * @dev Same as {xref-Address-functionCall-address-bytes-}[`functionCall`], but with
     * `errorMessage` as a fallback revert reason when `target` reverts.
     *
     * _Available since v3.1._
     */
    function functionCall(
        address target,
        bytes memory data,
        string memory errorMessage
    ) internal returns (bytes memory) {
        return functionCallWithValue(target, data, 0, errorMessage);
    }

    /**
     * @dev Same as {xref-Address-functionCall-address-bytes-}[`functionCall`],
     * but also transferring `value` wei to `target`.
     *
     * Requirements:
     *
     * - the calling contract must have an ETH balance of at least `value`.
     * - the called Solidity function must be `payable`.
     *
     * _Available since v3.1._
     */
    function functionCallWithValue(
        address target,
        bytes memory data,
        uint256 value
    ) internal returns (bytes memory) {
        return functionCallWithValue(target, data, value, "Address: low-level call with value failed");
    }

    /**
     * @dev Same as {xref-Address-functionCallWithValue-address-bytes-uint256-}[`functionCallWithValue`], but
     * with `errorMessage` as a fallback revert reason when `target` reverts.
     *
     * _Available since v3.1._
     */
    function functionCallWithValue(
        address target,
        bytes memory data,
        uint256 value,
        string memory errorMessage
    ) internal returns (bytes memory) {
        require(address(this).balance >= value, "Address: insufficient balance for call");
        (bool success, bytes memory returndata) = target.call{value: value}(data);
        return verifyCallResultFromTarget(target, success, returndata, errorMessage);
    }

    /**
     * @dev Same as {xref-Address-functionCall-address-bytes-}[`functionCall`],
     * but performing a static call.
     *
     * _Available since v3.3._
     */
    function functionStaticCall(address target, bytes memory data) internal view returns (bytes memory) {
        return functionStaticCall(target, data, "Address: low-level static call failed");
    }

    /**
     * @dev Same as {xref-Address-functionCall-address-bytes-string-}[`functionCall`],
     * but performing a static call.
     *
     * _Available since v3.3._
     */
    function functionStaticCall(
        address target,
        bytes memory data,
        string memory errorMessage
    ) internal view returns (bytes memory) {
        (bool success, bytes memory returndata) = target.staticcall(data);
        return verifyCallResultFromTarget(target, success, returndata, errorMessage);
    }

    /**
     * @dev Same as {xref-Address-functionCall-address-bytes-}[`functionCall`],
     * but performing a delegate call.
     *
     * _Available since v3.4._
     */
    function functionDelegateCall(address target, bytes memory data) internal returns (bytes memory) {
        return functionDelegateCall(target, data, "Address: low-level delegate call failed");
    }

    /**
     * @dev Same as {xref-Address-functionCall-address-bytes-string-}[`functionCall`],
     * but performing a delegate call.
     *
     * _Available since v3.4._
     */
    function functionDelegateCall(
        address target,
        bytes memory data,
        string memory errorMessage
    ) internal returns (bytes memory) {
        (bool success, bytes memory returndata) = target.delegatecall(data);
        return verifyCallResultFromTarget(target, success, returndata, errorMessage);
    }

    /**
     * @dev Tool to verify that a low level call to smart-contract was successful, and revert (either by bubbling
     * the revert reason or using the provided one) in case of unsuccessful call or if target was not a contract.
     *
     * _Available since v4.8._
     */
    function verifyCallResultFromTarget(
        address target,
        bool success,
        bytes memory returndata,
        string memory errorMessage
    ) internal view returns (bytes memory) {
        if (success) {
            if (returndata.length == 0) {
                // only check isContract if the call was successful and the return data is empty
                // otherwise we already know that it was a contract
                require(isContract(target), "Address: call to non-contract");
            }
            return returndata;
        } else {
            _revert(returndata, errorMessage);
        }
    }

    /**
     * @dev Tool to verify that a low level call was successful, and revert if it wasn't, either by bubbling the
     * revert reason or using the provided one.
     *
     * _Available since v4.3._
     */
    function verifyCallResult(
        bool success,
        bytes memory returndata,
        string memory errorMessage
    ) internal pure returns (bytes memory) {
        if (success) {
            return returndata;
        } else {
            _revert(returndata, errorMessage);
        }
    }

    function _revert(bytes memory returndata, string memory errorMessage) private pure {
        // Look for revert reason and bubble it up if present
        if (returndata.length > 0) {
            // The easiest way to bubble the revert reason is using memory via assembly
            /// @solidity memory-safe-assembly
            assembly {
                let returndata_size := mload(returndata)
                revert(add(32, returndata), returndata_size)
            }
        } else {
            revert(errorMessage);
        }
    }
}

/**
 * @title SafeERC20
 * @dev Wrappers around ERC20 operations that throw on failure (when the token
 * contract returns false). Tokens that return no value (and instead revert or
 * throw on failure) are also supported, non-reverting calls are assumed to be
 * successful.
 * To use this library you can add a `using SafeERC20 for IERC20;` statement to your contract,
 * which allows you to call the safe operations as `token.safeTransfer(...)`, etc.
 */
library SafeERC20 {
    using Address for address;

    function safeTransfer(
        IERC20 token,
        address to,
        uint256 value
    ) internal {
        _callOptionalReturn(token, abi.encodeWithSelector(token.transfer.selector, to, value));
    }

    function safeTransferFrom(
        IERC20 token,
        address from,
        address to,
        uint256 value
    ) internal {
        _callOptionalReturn(token, abi.encodeWithSelector(token.transferFrom.selector, from, to, value));
    }

    /**
     * @dev Deprecated. This function has issues similar to the ones found in
     * {IERC20-approve}, and its usage is discouraged.
     *
     * Whenever possible, use {safeIncreaseAllowance} and
     * {safeDecreaseAllowance} instead.
     */
    function safeApprove(
        IERC20 token,
        address spender,
        uint256 value
    ) internal {
        // safeApprove should only be called when setting an initial allowance,
        // or when resetting it to zero. To increase and decrease it, use
        // 'safeIncreaseAllowance' and 'safeDecreaseAllowance'
        require(
            (value == 0) || (token.allowance(address(this), spender) == 0),
            "SafeERC20: approve from non-zero to non-zero allowance"
        );
        _callOptionalReturn(token, abi.encodeWithSelector(token.approve.selector, spender, value));
    }

    function safeIncreaseAllowance(
        IERC20 token,
        address spender,
        uint256 value
    ) internal {
        uint256 newAllowance = token.allowance(address(this), spender) + value;
        _callOptionalReturn(token, abi.encodeWithSelector(token.approve.selector, spender, newAllowance));
    }

    function safeDecreaseAllowance(
        IERC20 token,
        address spender,
        uint256 value
    ) internal {
        unchecked {
            uint256 oldAllowance = token.allowance(address(this), spender);
            require(oldAllowance >= value, "SafeERC20: decreased allowance below zero");
            uint256 newAllowance = oldAllowance - value;
            _callOptionalReturn(token, abi.encodeWithSelector(token.approve.selector, spender, newAllowance));
        }
    }

    function safePermit(
        IERC20Permit token,
        address owner,
        address spender,
        uint256 value,
        uint256 deadline,
        uint8 v,
        bytes32 r,
        bytes32 s
    ) internal {
        uint256 nonceBefore = token.nonces(owner);
        token.permit(owner, spender, value, deadline, v, r, s);
        uint256 nonceAfter = token.nonces(owner);
        require(nonceAfter == nonceBefore + 1, "SafeERC20: permit did not succeed");
    }

    /**
     * @dev Imitates a Solidity high-level call (i.e. a regular function call to a contract), relaxing the requirement
     * on the return value: the return value is optional (but if data is returned, it must not be false).
     * @param token The token targeted by the call.
     * @param data The call data (encoded using abi.encode or one of its variants).
     */
    function _callOptionalReturn(IERC20 token, bytes memory data) private {
        // We need to perform a low level call here, to bypass Solidity's return data size checking mechanism, since
        // we're implementing it ourselves. We use {Address-functionCall} to perform this call, which verifies that
        // the target address contains contract code and also asserts for success in the low-level call.

        bytes memory returndata = address(token).functionCall(data, "SafeERC20: low-level call failed");
        if (returndata.length > 0) {
            // Return data is optional
            require(abi.decode(returndata, (bool)), "SafeERC20: ERC20 operation did not succeed");
        }
    }
}

// OpenZeppelin Contracts (last updated v4.8.0) (access/AccessControl.sol)

// OpenZeppelin Contracts v4.4.1 (access/IAccessControl.sol)

/**
 * @dev External interface of AccessControl declared to support ERC165 detection.
 */
interface IAccessControl {
    /**
     * @dev Emitted when `newAdminRole` is set as ``role``'s admin role, replacing `previousAdminRole`
     *
     * `DEFAULT_ADMIN_ROLE` is the starting admin for all roles, despite
     * {RoleAdminChanged} not being emitted signaling this.
     *
     * _Available since v3.1._
     */
    event RoleAdminChanged(bytes32 indexed role, bytes32 indexed previousAdminRole, bytes32 indexed newAdminRole);

    /**
     * @dev Emitted when `account` is granted `role`.
     *
     * `sender` is the account that originated the contract call, an admin role
     * bearer except when using {AccessControl-_setupRole}.
     */
    event RoleGranted(bytes32 indexed role, address indexed account, address indexed sender);

    /**
     * @dev Emitted when `account` is revoked `role`.
     *
     * `sender` is the account that originated the contract call:
     *   - if using `revokeRole`, it is the admin role bearer
     *   - if using `renounceRole`, it is the role bearer (i.e. `account`)
     */
    event RoleRevoked(bytes32 indexed role, address indexed account, address indexed sender);

    /**
     * @dev Returns `true` if `account` has been granted `role`.
     */
    function hasRole(bytes32 role, address account) external view returns (bool);

    /**
     * @dev Returns the admin role that controls `role`. See {grantRole} and
     * {revokeRole}.
     *
     * To change a role's admin, use {AccessControl-_setRoleAdmin}.
     */
    function getRoleAdmin(bytes32 role) external view returns (bytes32);

    /**
     * @dev Grants `role` to `account`.
     *
     * If `account` had not been already granted `role`, emits a {RoleGranted}
     * event.
     *
     * Requirements:
     *
     * - the caller must have ``role``'s admin role.
     */
    function grantRole(bytes32 role, address account) external;

    /**
     * @dev Revokes `role` from `account`.
     *
     * If `account` had been granted `role`, emits a {RoleRevoked} event.
     *
     * Requirements:
     *
     * - the caller must have ``role``'s admin role.
     */
    function revokeRole(bytes32 role, address account) external;

    /**
     * @dev Revokes `role` from the calling account.
     *
     * Roles are often managed via {grantRole} and {revokeRole}: this function's
     * purpose is to provide a mechanism for accounts to lose their privileges
     * if they are compromised (such as when a trusted device is misplaced).
     *
     * If the calling account had been granted `role`, emits a {RoleRevoked}
     * event.
     *
     * Requirements:
     *
     * - the caller must be `account`.
     */
    function renounceRole(bytes32 role, address account) external;
}

// OpenZeppelin Contracts v4.4.1 (utils/Context.sol)

/**
 * @dev Provides information about the current execution context, including the
 * sender of the transaction and its data. While these are generally available
 * via msg.sender and msg.data, they should not be accessed in such a direct
 * manner, since when dealing with meta-transactions the account sending and
 * paying for execution may not be the actual sender (as far as an application
 * is concerned).
 *
 * This contract is only required for intermediate, library-like contracts.
 */
abstract contract Context {
    function _msgSender() internal view virtual returns (address) {
        return msg.sender;
    }

    function _msgData() internal view virtual returns (bytes calldata) {
        return msg.data;
    }
}

// OpenZeppelin Contracts (last updated v4.8.0) (utils/Strings.sol)

// OpenZeppelin Contracts (last updated v4.8.0) (utils/math/Math.sol)

/**
 * @dev Standard math utilities missing in the Solidity language.
 */
library Math {
    enum Rounding {
        Down, // Toward negative infinity
        Up, // Toward infinity
        Zero // Toward zero
    }

    /**
     * @dev Returns the largest of two numbers.
     */
    function max(uint256 a, uint256 b) internal pure returns (uint256) {
        return a > b ? a : b;
    }

    /**
     * @dev Returns the smallest of two numbers.
     */
    function min(uint256 a, uint256 b) internal pure returns (uint256) {
        return a < b ? a : b;
    }

    /**
     * @dev Returns the average of two numbers. The result is rounded towards
     * zero.
     */
    function average(uint256 a, uint256 b) internal pure returns (uint256) {
        // (a + b) / 2 can overflow.
        return (a & b) + (a ^ b) / 2;
    }

    /**
     * @dev Returns the ceiling of the division of two numbers.
     *
     * This differs from standard division with `/` in that it rounds up instead
     * of rounding down.
     */
    function ceilDiv(uint256 a, uint256 b) internal pure returns (uint256) {
        // (a + b - 1) / b can overflow on addition, so we distribute.
        return a == 0 ? 0 : (a - 1) / b + 1;
    }

    /**
     * @notice Calculates floor(x * y / denominator) with full precision. Throws if result overflows a uint256 or denominator == 0
     * @dev Original credit to Remco Bloemen under MIT license (https://xn--2-umb.com/21/muldiv)
     * with further edits by Uniswap Labs also under MIT license.
     */
    function mulDiv(
        uint256 x,
        uint256 y,
        uint256 denominator
    ) internal pure returns (uint256 result) {
        unchecked {
            // 512-bit multiply [prod1 prod0] = x * y. Compute the product mod 2^256 and mod 2^256 - 1, then use
            // use the Chinese Remainder Theorem to reconstruct the 512 bit result. The result is stored in two 256
            // variables such that product = prod1 * 2^256 + prod0.
            uint256 prod0; // Least significant 256 bits of the product
            uint256 prod1; // Most significant 256 bits of the product
            assembly {
                let mm := mulmod(x, y, not(0))
                prod0 := mul(x, y)
                prod1 := sub(sub(mm, prod0), lt(mm, prod0))
            }

            // Handle non-overflow cases, 256 by 256 division.
            if (prod1 == 0) {
                return prod0 / denominator;
            }

            // Make sure the result is less than 2^256. Also prevents denominator == 0.
            require(denominator > prod1);

            ///////////////////////////////////////////////
            // 512 by 256 division.
            ///////////////////////////////////////////////

            // Make division exact by subtracting the remainder from [prod1 prod0].
            uint256 remainder;
            assembly {
                // Compute remainder using mulmod.
                remainder := mulmod(x, y, denominator)

                // Subtract 256 bit number from 512 bit number.
                prod1 := sub(prod1, gt(remainder, prod0))
                prod0 := sub(prod0, remainder)
            }

            // Factor powers of two out of denominator and compute largest power of two divisor of denominator. Always >= 1.
            // See https://cs.stackexchange.com/q/138556/92363.

            // Does not overflow because the denominator cannot be zero at this stage in the function.
            uint256 twos = denominator & (~denominator + 1);
            assembly {
                // Divide denominator by twos.
                denominator := div(denominator, twos)

                // Divide [prod1 prod0] by twos.
                prod0 := div(prod0, twos)

                // Flip twos such that it is 2^256 / twos. If twos is zero, then it becomes one.
                twos := add(div(sub(0, twos), twos), 1)
            }

            // Shift in bits from prod1 into prod0.
            prod0 |= prod1 * twos;

            // Invert denominator mod 2^256. Now that denominator is an odd number, it has an inverse modulo 2^256 such
            // that denominator * inv = 1 mod 2^256. Compute the inverse by starting with a seed that is correct for
            // four bits. That is, denominator * inv = 1 mod 2^4.
            uint256 inverse = (3 * denominator) ^ 2;

            // Use the Newton-Raphson iteration to improve the precision. Thanks to Hensel's lifting lemma, this also works
            // in modular arithmetic, doubling the correct bits in each step.
            inverse *= 2 - denominator * inverse; // inverse mod 2^8
            inverse *= 2 - denominator * inverse; // inverse mod 2^16
            inverse *= 2 - denominator * inverse; // inverse mod 2^32
            inverse *= 2 - denominator * inverse; // inverse mod 2^64
            inverse *= 2 - denominator * inverse; // inverse mod 2^128
            inverse *= 2 - denominator * inverse; // inverse mod 2^256

            // Because the division is now exact we can divide by multiplying with the modular inverse of denominator.
            // This will give us the correct result modulo 2^256. Since the preconditions guarantee that the outcome is
            // less than 2^256, this is the final result. We don't need to compute the high bits of the result and prod1
            // is no longer required.
            result = prod0 * inverse;
            return result;
        }
    }

    /**
     * @notice Calculates x * y / denominator with full precision, following the selected rounding direction.
     */
    function mulDiv(
        uint256 x,
        uint256 y,
        uint256 denominator,
        Rounding rounding
    ) internal pure returns (uint256) {
        uint256 result = mulDiv(x, y, denominator);
        if (rounding == Rounding.Up && mulmod(x, y, denominator) > 0) {
            result += 1;
        }
        return result;
    }

    /**
     * @dev Returns the square root of a number. If the number is not a perfect square, the value is rounded down.
     *
     * Inspired by Henry S. Warren, Jr.'s "Hacker's Delight" (Chapter 11).
     */
    function sqrt(uint256 a) internal pure returns (uint256) {
        if (a == 0) {
            return 0;
        }

        // For our first guess, we get the biggest power of 2 which is smaller than the square root of the target.
        //
        // We know that the "msb" (most significant bit) of our target number `a` is a power of 2 such that we have
        // `msb(a) <= a < 2*msb(a)`. This value can be written `msb(a)=2**k` with `k=log2(a)`.
        //
        // This can be rewritten `2**log2(a) <= a < 2**(log2(a) + 1)`
        // → `sqrt(2**k) <= sqrt(a) < sqrt(2**(k+1))`
        // → `2**(k/2) <= sqrt(a) < 2**((k+1)/2) <= 2**(k/2 + 1)`
        //
        // Consequently, `2**(log2(a) / 2)` is a good first approximation of `sqrt(a)` with at least 1 correct bit.
        uint256 result = 1 << (log2(a) >> 1);

        // At this point `result` is an estimation with one bit of precision. We know the true value is a uint128,
        // since it is the square root of a uint256. Newton's method converges quadratically (precision doubles at
        // every iteration). We thus need at most 7 iteration to turn our partial result with one bit of precision
        // into the expected uint128 result.
        unchecked {
            result = (result + a / result) >> 1;
            result = (result + a / result) >> 1;
            result = (result + a / result) >> 1;
            result = (result + a / result) >> 1;
            result = (result + a / result) >> 1;
            result = (result + a / result) >> 1;
            result = (result + a / result) >> 1;
            return min(result, a / result);
        }
    }

    /**
     * @notice Calculates sqrt(a), following the selected rounding direction.
     */
    function sqrt(uint256 a, Rounding rounding) internal pure returns (uint256) {
        unchecked {
            uint256 result = sqrt(a);
            return result + (rounding == Rounding.Up && result * result < a ? 1 : 0);
        }
    }

    /**
     * @dev Return the log in base 2, rounded down, of a positive value.
     * Returns 0 if given 0.
     */
    function log2(uint256 value) internal pure returns (uint256) {
        uint256 result = 0;
        unchecked {
            if (value >> 128 > 0) {
                value >>= 128;
                result += 128;
            }
            if (value >> 64 > 0) {
                value >>= 64;
                result += 64;
            }
            if (value >> 32 > 0) {
                value >>= 32;
                result += 32;
            }
            if (value >> 16 > 0) {
                value >>= 16;
                result += 16;
            }
            if (value >> 8 > 0) {
                value >>= 8;
                result += 8;
            }
            if (value >> 4 > 0) {
                value >>= 4;
                result += 4;
            }
            if (value >> 2 > 0) {
                value >>= 2;
                result += 2;
            }
            if (value >> 1 > 0) {
                result += 1;
            }
        }
        return result;
    }

    /**
     * @dev Return the log in base 2, following the selected rounding direction, of a positive value.
     * Returns 0 if given 0.
     */
    function log2(uint256 value, Rounding rounding) internal pure returns (uint256) {
        unchecked {
            uint256 result = log2(value);
            return result + (rounding == Rounding.Up && 1 << result < value ? 1 : 0);
        }
    }

    /**
     * @dev Return the log in base 10, rounded down, of a positive value.
     * Returns 0 if given 0.
     */
    function log10(uint256 value) internal pure returns (uint256) {
        uint256 result = 0;
        unchecked {
            if (value >= 10**64) {
                value /= 10**64;
                result += 64;
            }
            if (value >= 10**32) {
                value /= 10**32;
                result += 32;
            }
            if (value >= 10**16) {
                value /= 10**16;
                result += 16;
            }
            if (value >= 10**8) {
                value /= 10**8;
                result += 8;
            }
            if (value >= 10**4) {
                value /= 10**4;
                result += 4;
            }
            if (value >= 10**2) {
                value /= 10**2;
                result += 2;
            }
            if (value >= 10**1) {
                result += 1;
            }
        }
        return result;
    }

    /**
     * @dev Return the log in base 10, following the selected rounding direction, of a positive value.
     * Returns 0 if given 0.
     */
    function log10(uint256 value, Rounding rounding) internal pure returns (uint256) {
        unchecked {
            uint256 result = log10(value);
            return result + (rounding == Rounding.Up && 10**result < value ? 1 : 0);
        }
    }

    /**
     * @dev Return the log in base 256, rounded down, of a positive value.
     * Returns 0 if given 0.
     *
     * Adding one to the result gives the number of pairs of hex symbols needed to represent `value` as a hex string.
     */
    function log256(uint256 value) internal pure returns (uint256) {
        uint256 result = 0;
        unchecked {
            if (value >> 128 > 0) {
                value >>= 128;
                result += 16;
            }
            if (value >> 64 > 0) {
                value >>= 64;
                result += 8;
            }
            if (value >> 32 > 0) {
                value >>= 32;
                result += 4;
            }
            if (value >> 16 > 0) {
                value >>= 16;
                result += 2;
            }
            if (value >> 8 > 0) {
                result += 1;
            }
        }
        return result;
    }

    /**
     * @dev Return the log in base 10, following the selected rounding direction, of a positive value.
     * Returns 0 if given 0.
     */
    function log256(uint256 value, Rounding rounding) internal pure returns (uint256) {
        unchecked {
            uint256 result = log256(value);
            return result + (rounding == Rounding.Up && 1 << (result * 8) < value ? 1 : 0);
        }
    }
}

/**
 * @dev String operations.
 */
library Strings {
    bytes16 private constant _SYMBOLS = "0123456789abcdef";
    uint8 private constant _ADDRESS_LENGTH = 20;

    /**
     * @dev Converts a `uint256` to its ASCII `string` decimal representation.
     */
    function toString(uint256 value) internal pure returns (string memory) {
        unchecked {
            uint256 length = Math.log10(value) + 1;
            string memory buffer = new string(length);
            uint256 ptr;
            /// @solidity memory-safe-assembly
            assembly {
                ptr := add(buffer, add(32, length))
            }
            while (true) {
                ptr--;
                /// @solidity memory-safe-assembly
                assembly {
                    mstore8(ptr, byte(mod(value, 10), _SYMBOLS))
                }
                value /= 10;
                if (value == 0) break;
            }
            return buffer;
        }
    }

    /**
     * @dev Converts a `uint256` to its ASCII `string` hexadecimal representation.
     */
    function toHexString(uint256 value) internal pure returns (string memory) {
        unchecked {
            return toHexString(value, Math.log256(value) + 1);
        }
    }

    /**
     * @dev Converts a `uint256` to its ASCII `string` hexadecimal representation with fixed length.
     */
    function toHexString(uint256 value, uint256 length) internal pure returns (string memory) {
        bytes memory buffer = new bytes(2 * length + 2);
        buffer[0] = "0";
        buffer[1] = "x";
        for (uint256 i = 2 * length + 1; i > 1; --i) {
            buffer[i] = _SYMBOLS[value & 0xf];
            value >>= 4;
        }
        require(value == 0, "Strings: hex length insufficient");
        return string(buffer);
    }

    /**
     * @dev Converts an `address` with fixed length of 20 bytes to its not checksummed ASCII `string` hexadecimal representation.
     */
    function toHexString(address addr) internal pure returns (string memory) {
        return toHexString(uint256(uint160(addr)), _ADDRESS_LENGTH);
    }
}

// OpenZeppelin Contracts v4.4.1 (utils/introspection/ERC165.sol)

// OpenZeppelin Contracts v4.4.1 (utils/introspection/IERC165.sol)

/**
 * @dev Interface of the ERC165 standard, as defined in the
 * https://eips.ethereum.org/EIPS/eip-165[EIP].
 *
 * Implementers can declare support of contract interfaces, which can then be
 * queried by others ({ERC165Checker}).
 *
 * For an implementation, see {ERC165}.
 */
interface IERC165 {
    /**
     * @dev Returns true if this contract implements the interface defined by
     * `interfaceId`. See the corresponding
     * https://eips.ethereum.org/EIPS/eip-165#how-interfaces-are-identified[EIP section]
     * to learn more about how these ids are created.
     *
     * This function call must use less than 30 000 gas.
     */
    function supportsInterface(bytes4 interfaceId) external view returns (bool);
}

/**
 * @dev Implementation of the {IERC165} interface.
 *
 * Contracts that want to implement ERC165 should inherit from this contract and override {supportsInterface} to check
 * for the additional interface id that will be supported. For example:
 *
 * ```solidity
 * function supportsInterface(bytes4 interfaceId) public view virtual override returns (bool) {
 *     return interfaceId == type(MyInterface).interfaceId || super.supportsInterface(interfaceId);
 * }
 * ```
 *
 * Alternatively, {ERC165Storage} provides an easier to use but more expensive implementation.
 */
abstract contract ERC165 is IERC165 {
    /**
     * @dev See {IERC165-supportsInterface}.
     */
    function supportsInterface(bytes4 interfaceId) public view virtual override returns (bool) {
        return interfaceId == type(IERC165).interfaceId;
    }
}

/**
 * @dev Contract module that allows children to implement role-based access
 * control mechanisms. This is a lightweight version that doesn't allow enumerating role
 * members except through off-chain means by accessing the contract event logs. Some
 * applications may benefit from on-chain enumerability, for those cases see
 * {AccessControlEnumerable}.
 *
 * Roles are referred to by their `bytes32` identifier. These should be exposed
 * in the external API and be unique. The best way to achieve this is by
 * using `public constant` hash digests:
 *
 * ```
 * bytes32 public constant MY_ROLE = keccak256("MY_ROLE");
 * ```
 *
 * Roles can be used to represent a set of permissions. To restrict access to a
 * function call, use {hasRole}:
 *
 * ```
 * function foo() public {
 *     require(hasRole(MY_ROLE, msg.sender));
 *     ...
 * }
 * ```
 *
 * Roles can be granted and revoked dynamically via the {grantRole} and
 * {revokeRole} functions. Each role has an associated admin role, and only
 * accounts that have a role's admin role can call {grantRole} and {revokeRole}.
 *
 * By default, the admin role for all roles is `DEFAULT_ADMIN_ROLE`, which means
 * that only accounts with this role will be able to grant or revoke other
 * roles. More complex role relationships can be created by using
 * {_setRoleAdmin}.
 *
 * WARNING: The `DEFAULT_ADMIN_ROLE` is also its own admin: it has permission to
 * grant and revoke this role. Extra precautions should be taken to secure
 * accounts that have been granted it.
 */
abstract contract AccessControl is Context, IAccessControl, ERC165 {
    struct RoleData {
        mapping(address => bool) members;
        bytes32 adminRole;
    }

    mapping(bytes32 => RoleData) private _roles;

    bytes32 public constant DEFAULT_ADMIN_ROLE = 0x00;

    /**
     * @dev Modifier that checks that an account has a specific role. Reverts
     * with a standardized message including the required role.
     *
     * The format of the revert reason is given by the following regular expression:
     *
     *  /^AccessControl: account (0x[0-9a-f]{40}) is missing role (0x[0-9a-f]{64})$/
     *
     * _Available since v4.1._
     */
    modifier onlyRole(bytes32 role) {
        _checkRole(role);
        _;
    }

    /**
     * @dev See {IERC165-supportsInterface}.
     */
    function supportsInterface(bytes4 interfaceId) public view virtual override returns (bool) {
        return interfaceId == type(IAccessControl).interfaceId || super.supportsInterface(interfaceId);
    }

    /**
     * @dev Returns `true` if `account` has been granted `role`.
     */
    function hasRole(bytes32 role, address account) public view virtual override returns (bool) {
        return _roles[role].members[account];
    }

    /**
     * @dev Revert with a standard message if `_msgSender()` is missing `role`.
     * Overriding this function changes the behavior of the {onlyRole} modifier.
     *
     * Format of the revert message is described in {_checkRole}.
     *
     * _Available since v4.6._
     */
    function _checkRole(bytes32 role) internal view virtual {
        _checkRole(role, _msgSender());
    }

    /**
     * @dev Revert with a standard message if `account` is missing `role`.
     *
     * The format of the revert reason is given by the following regular expression:
     *
     *  /^AccessControl: account (0x[0-9a-f]{40}) is missing role (0x[0-9a-f]{64})$/
     */
    function _checkRole(bytes32 role, address account) internal view virtual {
        if (!hasRole(role, account)) {
            revert(
                string(
                    abi.encodePacked(
                        "AccessControl: account ",
                        Strings.toHexString(account),
                        " is missing role ",
                        Strings.toHexString(uint256(role), 32)
                    )
                )
            );
        }
    }

    /**
     * @dev Returns the admin role that controls `role`. See {grantRole} and
     * {revokeRole}.
     *
     * To change a role's admin, use {_setRoleAdmin}.
     */
    function getRoleAdmin(bytes32 role) public view virtual override returns (bytes32) {
        return _roles[role].adminRole;
    }

    /**
     * @dev Grants `role` to `account`.
     *
     * If `account` had not been already granted `role`, emits a {RoleGranted}
     * event.
     *
     * Requirements:
     *
     * - the caller must have ``role``'s admin role.
     *
     * May emit a {RoleGranted} event.
     */
    function grantRole(bytes32 role, address account) public virtual override onlyRole(getRoleAdmin(role)) {
        _grantRole(role, account);
    }

    /**
     * @dev Revokes `role` from `account`.
     *
     * If `account` had been granted `role`, emits a {RoleRevoked} event.
     *
     * Requirements:
     *
     * - the caller must have ``role``'s admin role.
     *
     * May emit a {RoleRevoked} event.
     */
    function revokeRole(bytes32 role, address account) public virtual override onlyRole(getRoleAdmin(role)) {
        _revokeRole(role, account);
    }

    /**
     * @dev Revokes `role` from the calling account.
     *
     * Roles are often managed via {grantRole} and {revokeRole}: this function's
     * purpose is to provide a mechanism for accounts to lose their privileges
     * if they are compromised (such as when a trusted device is misplaced).
     *
     * If the calling account had been revoked `role`, emits a {RoleRevoked}
     * event.
     *
     * Requirements:
     *
     * - the caller must be `account`.
     *
     * May emit a {RoleRevoked} event.
     */
    function renounceRole(bytes32 role, address account) public virtual override {
        require(account == _msgSender(), "AccessControl: can only renounce roles for self");

        _revokeRole(role, account);
    }

    /**
     * @dev Grants `role` to `account`.
     *
     * If `account` had not been already granted `role`, emits a {RoleGranted}
     * event. Note that unlike {grantRole}, this function doesn't perform any
     * checks on the calling account.
     *
     * May emit a {RoleGranted} event.
     *
     * [WARNING]
     * ====
     * This function should only be called from the constructor when setting
     * up the initial roles for the system.
     *
     * Using this function in any other way is effectively circumventing the admin
     * system imposed by {AccessControl}.
     * ====
     *
     * NOTE: This function is deprecated in favor of {_grantRole}.
     */
    function _setupRole(bytes32 role, address account) internal virtual {
        _grantRole(role, account);
    }

    /**
     * @dev Sets `adminRole` as ``role``'s admin role.
     *
     * Emits a {RoleAdminChanged} event.
     */
    function _setRoleAdmin(bytes32 role, bytes32 adminRole) internal virtual {
        bytes32 previousAdminRole = getRoleAdmin(role);
        _roles[role].adminRole = adminRole;
        emit RoleAdminChanged(role, previousAdminRole, adminRole);
    }

    /**
     * @dev Grants `role` to `account`.
     *
     * Internal function without access restriction.
     *
     * May emit a {RoleGranted} event.
     */
    function _grantRole(bytes32 role, address account) internal virtual {
        if (!hasRole(role, account)) {
            _roles[role].members[account] = true;
            emit RoleGranted(role, account, _msgSender());
        }
    }

    /**
     * @dev Revokes `role` from `account`.
     *
     * Internal function without access restriction.
     *
     * May emit a {RoleRevoked} event.
     */
    function _revokeRole(bytes32 role, address account) internal virtual {
        if (hasRole(role, account)) {
            _roles[role].members[account] = false;
            emit RoleRevoked(role, account, _msgSender());
        }
    }
}

interface ICurvePool {
    // function get_balances() external view returns (uint256[] memory);

    function balances(uint256 i) external view returns (uint256);

    function remove_liquidity_one_coin(
        uint256 burn_amount,
        int128 coin_idx,
        uint256 min_received
    ) external returns (uint256);

    function remove_liquidity(
        uint256 burn_amount,
        uint256[2] memory amounts
    ) external returns (uint256[2] memory);

    function add_liquidity(
        uint256[2] memory amounts,
        uint256 min_mint_amount
    ) external returns (uint256);

    function calc_token_amount(
        uint256[2] memory _amounts,
        bool _is_deposit
    ) external view returns (uint256);

    function exchange(
        int128 i,
        int128 j,
        uint256 dx,
        uint256 min_dy
    ) external returns (uint256);

    function get_virtual_price() external view returns (uint256);
}

interface IXETH is IERC20 {
    function burnShares(uint256 amount) external;

    function mintShares(uint256 amount) external;
}

// OpenZeppelin Contracts (last updated v4.7.0) (access/Ownable.sol)

/**
 * @dev Contract module which provides a basic access control mechanism, where
 * there is an account (an owner) that can be granted exclusive access to
 * specific functions.
 *
 * By default, the owner account will be the one that deploys the contract. This
 * can later be changed with {transferOwnership}.
 *
 * This module is used through inheritance. It will make available the modifier
 * `onlyOwner`, which can be applied to your functions to restrict their use to
 * the owner.
 */
abstract contract Ownable is Context {
    address private _owner;

    event OwnershipTransferred(address indexed previousOwner, address indexed newOwner);

    /**
     * @dev Initializes the contract setting the deployer as the initial owner.
     */
    constructor() {
        _transferOwnership(_msgSender());
    }

    /**
     * @dev Throws if called by any account other than the owner.
     */
    modifier onlyOwner() {
        _checkOwner();
        _;
    }

    /**
     * @dev Returns the address of the current owner.
     */
    function owner() public view virtual returns (address) {
        return _owner;
    }

    /**
     * @dev Throws if the sender is not the owner.
     */
    function _checkOwner() internal view virtual {
        require(owner() == _msgSender(), "Ownable: caller is not the owner");
    }

    /**
     * @dev Leaves the contract without owner. It will not be possible to call
     * `onlyOwner` functions anymore. Can only be called by the current owner.
     *
     * NOTE: Renouncing ownership will leave the contract without an owner,
     * thereby removing any functionality that is only available to the owner.
     */
    function renounceOwnership() public virtual onlyOwner {
        _transferOwnership(address(0));
    }

    /**
     * @dev Transfers ownership of the contract to a new account (`newOwner`).
     * Can only be called by the current owner.
     */
    function transferOwnership(address newOwner) public virtual onlyOwner {
        require(newOwner != address(0), "Ownable: new owner is the zero address");
        _transferOwnership(newOwner);
    }

    /**
     * @dev Transfers ownership of the contract to a new account (`newOwner`).
     * Internal function without access restriction.
     */
    function _transferOwnership(address newOwner) internal virtual {
        address oldOwner = _owner;
        _owner = newOwner;
        emit OwnershipTransferred(oldOwner, newOwner);
    }
}

interface ICVXBooster {
    struct PoolInfo {
        address lptoken;
        address token;
        address gauge;
        address crvRewards;
        address stash;
        bool shutdown;
    }

    function poolInfo(uint256 _pid) external view returns (PoolInfo memory);

    function depositAll(uint256 _pid, bool _stake) external returns (bool);

    function withdrawAll(uint256 _pid) external returns (bool);

    function deposit(
        uint256 _pid,
        uint256 _amount,
        bool _stake
    ) external returns (bool);

    function withdraw(uint256 _pid, uint256 _amount) external returns (bool);
}

interface IBaseRewardPool {
    function withdrawAndUnwrap(
        uint256 amount,
        bool claim
    ) external returns (bool);

    function withdrawAll(bool claim) external;

    function withdrawAllAndUnwrap(bool claim) external;

    function withdraw(uint256 amount, bool claim) external;

    function stakeFor(address _for, uint256 _amount) external returns (bool);

    function stakeAll() external returns (bool);

    function stake(uint256 _amount) external returns (bool);

    function earned(address account) external view returns (uint256);

    function getReward(
        address _account,
        bool _claimExtras
    ) external returns (bool);

    function getReward() external returns (bool);

    function balanceOf(address account) external view returns (uint256);

    function rewardToken() external view returns (address);
}

contract CVXStaker is Ownable {
    using SafeERC20 for IERC20;

    address public operator;
    // @notice CLP tokens for curve pool
    IERC20 public immutable clpToken;
    CvxPoolInfo public cvxPoolInfo;
    // @notice Cvx booster
    ICVXBooster public immutable booster;

    address public rewardsRecipient;
    address[] public rewardTokens;

    struct CvxPoolInfo {
        address token;
        address rewards;
        uint32 pId;
    }

    struct Position {
        uint256 staked;
        uint256 earned;
    }

    error NotOperator();
    error NotOperatorOrOwner();

    event SetCvxPoolInfo(uint32 indexed pId, address token, address rewards);
    event SetOperator(address operator);
    event RecoveredToken(address token, address to, uint256 amount);
    event SetRewardsRecipient(address recipient);
    event SetRewardTokens(address[] newTokens);

    constructor(
        address _operator,
        IERC20 _clpToken,
        ICVXBooster _booster,
        address[] memory _rewardTokens
    ) {
        operator = _operator;
        clpToken = _clpToken;
        booster = _booster;
        rewardTokens = _rewardTokens;
    }

    /**
     * @dev Sets the CVX pool information.
     * @param _pId The pool ID of the CVX pool.
     * @param _token The address of the CLP token.
     * @param _rewards The address of the CVX reward pool.
     * Only the contract owner can call this function.
     */
    function setCvxPoolInfo(
        uint32 _pId,
        address _token,
        address _rewards
    ) external onlyOwner {
        cvxPoolInfo.pId = _pId;
        cvxPoolInfo.token = _token;
        cvxPoolInfo.rewards = _rewards;

        emit SetCvxPoolInfo(_pId, _token, _rewards);
    }

    /**
     * @notice Set operator
     * @param _operator New operator
     */
    function setOperator(address _operator) external onlyOwner {
        operator = _operator;

        emit SetOperator(_operator);
    }

    /**
     * @dev Sets the address of the rewards recipient.
     * @param _recipeint The address of the rewards recipient.
     * Only the contract owner can call this function.
     */
    function setRewardsRecipient(address _recipeint) external onlyOwner {
        rewardsRecipient = _recipeint;

        emit SetRewardsRecipient(_recipeint);
    }

    function setRewardTokens(address[] calldata newTokens) external onlyOwner {
      rewardTokens = newTokens;

      emit SetRewardTokens(newTokens);
    }

    /**
     * @notice Recover any token from cvxStaker 
     * @param token Token to recover
     * @param to Recipient address
     * @param amount Amount to recover
     */
    function recoverToken(
        address token,
        address to,
        uint256 amount
    ) external onlyOwner {
        IERC20(token).safeTransfer(to, amount);

        emit RecoveredToken(token, to, amount);
    }

    /**
     * @dev Checks whether the CVX pool is currently shutdown.
     * @return A boolean indicating whether the CVX pool is currently shutdown.
     */
    function isCvxShutdown() public view returns (bool) {
        // It's not necessary to check that the booster itself is shutdown, as that can only
        // be shutdown once all the pools are shutdown - see Cvx BoosterOwner.shutdownSystem()
        return booster.poolInfo(cvxPoolInfo.pId).shutdown;
    }

    /**
     * @dev Deposits a specified amount of CLP tokens into the booster and stakes them in the reward pool.
     * @param amount The amount of CLP tokens to deposit and stake.
     * Only the operator can call this function.
     */
    function depositAndStake(uint256 amount) external onlyOperator {
        // Only deposit if the aura pool is open. Otherwise leave the CLP Token in this contract.
        if (!isCvxShutdown()) {
            clpToken.safeIncreaseAllowance(address(booster), amount);
            booster.deposit(cvxPoolInfo.pId, amount, true);
        }
    }

    /**
     * @dev Withdraws a specified amount of staked tokens from the reward pool and unwraps them to the original tokens.
     * @param amount The amount of tokens to withdraw and unwrap.
     * @param claim A boolean indicating whether to claim rewards before withdrawing.
     * @param to The address to receive the unwrapped tokens.
     * If set to 0x0, the tokens will remain in the contract.
     * Only the contract owner or operator can call this function.
     */
    function withdrawAndUnwrap(
        uint256 amount,
        bool claim,
        address to
    ) external onlyOperatorOrOwner {
        // Optimistically use CLP balance in this contract, and then try and unstake any remaining
        uint256 clpBalance = clpToken.balanceOf(address(this));
        uint256 toUnstake = (amount < clpBalance) ? 0 : amount - clpBalance;
        if (toUnstake > 0) {
            IBaseRewardPool(cvxPoolInfo.rewards).withdrawAndUnwrap(
                toUnstake,
                claim
            );
        }

        if (to != address(0)) {
            // unwrapped amount is 1 to 1
            clpToken.safeTransfer(to, amount);
        }
    }

    /**
     * @dev Withdraws all staked tokens from the reward pool and unwraps them to the original tokens.
     * @param claim A boolean indicating whether to claim rewards before withdrawing.
     * @param sendToOwner A boolean indicating whether to send the unwrapped tokens to the owner.
     * If false, the tokens will remain in the contract.
     * Only the contract owner can call this function.
     */
    function withdrawAllAndUnwrap(
        bool claim,
        bool sendToOwner
    ) external onlyOwner {
        IBaseRewardPool(cvxPoolInfo.rewards).withdrawAllAndUnwrap(claim);
        if (sendToOwner) {
            uint256 totalBalance = clpToken.balanceOf(address(this));
            /// @dev msg.sender is the owner, due to onlyOwner modifier
            clpToken.safeTransfer(msg.sender, totalBalance);
        }
    }

    /**
     * @dev Claims the rewards and transfers them to the rewards recipient, if specified.
     * @param claimExtras A boolean indicating whether to claim extra rewards.
     */
    function getReward(bool claimExtras) external {
        IBaseRewardPool(cvxPoolInfo.rewards).getReward(
            address(this),
            claimExtras
        );
    }

    error OutOfBounds(uint8 check);

    function transferReward(uint256 initialIndex, uint256 lastIndex) external {
      if (initialIndex >= lastIndex) {
        revert OutOfBounds(0);
      }
      if (lastIndex > rewardTokens.length) {
        revert OutOfBounds(1);
      }

      if (rewardsRecipient != address(0)) {
        for (uint i = initialIndex; i < lastIndex; ) {
            uint256 balance = IERC20(rewardTokens[i]).balanceOf(
                address(this)
            );
            if (balance != 0) {
              IERC20(rewardTokens[i]).safeTransfer(rewardsRecipient, balance);
            }
            unchecked {++i;}
        }
      }
    }

    /**
     * @dev Returns the current staked balance of the contract.
     * @return balance The current staked balance.
     */
    function stakedBalance() public view returns (uint256 balance) {
        balance = IBaseRewardPool(cvxPoolInfo.rewards).balanceOf(address(this));
    }

    function getTotalBalance() public view returns(uint256 balance) {
      unchecked {
        balance = stakedBalance() + clpToken.balanceOf(address(this));
      }
    }

    /**
     * @dev Returns the amount of earned rewards by the contract.
     * @return earnedRewards The amount of earned rewards.
     */
    function earned() public view returns (uint256 earnedRewards) {
        earnedRewards = IBaseRewardPool(cvxPoolInfo.rewards).earned(
            address(this)
        );
    }

    /**
     * @notice show staked position and earned rewards
     */
    function showPositions() external view returns (Position memory position) {
        position.staked = stakedBalance();
        position.earned = earned();
    }

    /// @dev Modifier to restrict function execution to only the contract operator.
    /// @notice Throws a custom exception `NotOperator` if the caller is not the operator.
    modifier onlyOperator() {
        if (msg.sender != operator) {
            revert NotOperator();
        }
        _;
    }

    /// @dev Modifier to restrict function execution to only the contract operator or owner.
    /// @notice Throws a custom exception `NotOperatorOrOwner` if the caller is neither the operator nor the owner.
    modifier onlyOperatorOrOwner() {
        if (msg.sender != operator && msg.sender != owner()) {
            revert NotOperatorOrOwner();
        }
        _;
    }
}

contract xETH_AMO is AccessControl {
    using SafeERC20 for IERC20;

    /// @notice Thrown when the xETH-stETH LP balance of the AMO is too low for the rebalancing operation.
    error LpBalanceTooLow();

    /// @notice Thrown when either the stETH or xETH balance in the pool is zero, which would prevent rebalancing.
    error ZeroBalancePool();

    /// @notice Thrown when a zero address is provided as an input, which is not allowed.
    error ZeroAddressProvided();

    /// @notice Thrown when a function is called with a zero value, which is not allowed.
    error ZeroValueProvided();

    /// @notice Thrown when the setSlippage values are invalid
    error InvalidSetSlippage();

    /// @notice Thrown when a rebalance attempt is made before the cooldown period has finished.
    error CooldownNotFinished();

    /// @notice Thrown when a rebalance attempt is made, but the current pool ratios do not require rebalancing.
    error RebalanceNotRequired();

    /// @notice Thrown when a rebalanceUp operation is not allowed based on the current pool ratios.
    error RebalanceUpNotAllowed();

    /// @notice Thrown when a rebalanceDown operation is not allowed based on the current pool ratios.
    error RebalanceDownNotAllowed();

    /// @notice Thrown when the requested rebalanceUp operation exceeds the allowed rebalanceUpCap.
    error RebalanceUpCapExceeded();

    /// @notice Thrown when the requested rebalanceDown operation exceeds the allowed rebalanceDownCap.
    error RebalanceDownCapExceeded();

    /// @notice Emitted when a rebalanceUp operation is performed.
    /// @param quote The chosen quote for rebalancing.
    /// @param xETHamountReceived The actual amount of xETH received after burning LP tokens.
    event RebalanceUpFinished(
        RebalanceUpQuote quote,
        uint256 xETHamountReceived
    );

    /// @notice Emitted when a rebalanceDown operation is performed.
    /// @param quote The chosen quote for rebalancing.
    /// @param lpAmountReceived The actual amount of xETH-stETH LP tokens received after minting xETH.
    event RebalanceDownFinished(
        RebalanceDownQuote quote,
        uint256 lpAmountReceived
    );

    /// @notice Emitted when the defender address is updated.
    /// @param oldDefender The previous defender address.
    /// @param newDefender The new defender address.
    event DefenderUpdated(address oldDefender, address newDefender);

    /// @notice Emitted when the upSlippage and downSlippage parameters are updated
    /// @param oldUpSlippage The old slippage value for rebalanceUp.
    /// @param newUpSlippage The new slippage value for rebalanceUp.
    /// @param oldDownSlippage The old slippage value for rebalanceDown.
    /// @param newDownSlippage The new slippage value for rebalanceDown.
    event SlippageUpdated(
      uint256 oldUpSlippage,
      uint256 newUpSlippage,
      uint256 oldDownSlippage,
      uint256 newDownSlippage
    );

    /// @notice Emitted when the rebalanceUpCap is updated.
    /// @param oldRebalanceUpCap The previous rebalanceUpCap value.
    /// @param newRebalanceUpCap The new rebalanceUpCap value.
    event RebalanceUpCapUpdated(
        uint256 oldRebalanceUpCap,
        uint256 newRebalanceUpCap
    );

    /// @notice Emitted when the rebalanceDownCap is updated.
    /// @param oldRebalanceDownCap The previous rebalanceDownCap value.
    /// @param newRebalanceDownCap The new rebalanceDownCap value.
    event RebalanceDownCapUpdated(
        uint256 oldRebalanceDownCap,
        uint256 newRebalanceDownCap
    );

    /// @notice Emitted when the cooldownBlocks is updated.
    /// @param oldCooldownBlocks The previous cooldownBlocks value.
    /// @param newCooldownBlocks The new cooldownBlocks value.
    event CooldownBlocksUpdated(
        uint256 oldCooldownBlocks,
        uint256 newCooldownBlocks
    );

    /// @notice Emitted when the CVXStaker address is updated.
    /// @param oldCVXStaker The previous CVXStaker address.
    /// @param newCVXStaker The new CVXStaker address.
    event CVXStakerUpdated(address oldCVXStaker, address newCVXStaker);

    /// @notice Emitted when the rebalance up threshold is set.
    /// @param oldThreshold The old rebalance up threshold.
    /// @param newThreshold The new rebalance up threshold.
    event SetRebalanceUpThreshold(uint256 oldThreshold, uint256 newThreshold);

    /// @notice Emitted when the rebalance down threshold is set.
    /// @param oldThreshold The old rebalance down threshold.
    /// @param newThreshold The new rebalance down threshold.
    event SetRebalanceDownThreshold(uint256 oldThreshold, uint256 newThreshold);

    event RecoveredToken(address token, address to, uint256 amount);

    /// @dev REBALANCE_DEFENDER_ROLE is the role that allows the defender to call rebalance()
    bytes32 public constant REBALANCE_DEFENDER_ROLE =
        keccak256("REBALANCE_DEFENDER_ROLE");

    /// @dev BASE_UNIT is the base unit used for calculations (1E18)
    uint256 public constant BASE_UNIT = 1E18;

    /// @dev xETHIndex is the index of xETH in the Curve pool
    uint256 public immutable xETHIndex;

    /// @dev stETHIndex is the index of stETH in the Curve pool
    uint256 public immutable stETHIndex;

    /// @dev xETH is the xETH token contract
    IXETH public immutable xETH;

    /// @dev stETH is the stETH token contract
    IERC20 public immutable stETH;

    /// @dev curvePool is the Curve pool contract
    ICurvePool public immutable curvePool;

    /// @dev upSlippage is the maximum slippage allowed when rebalancing up
    /// @notice 1E14 = 1 BPS
    uint256 public upSlippage;
    
    /// @dev downSlippage is the maximum slippage allowed when rebalancing down 
    /// @notice 1E14 = 1 BPS
    uint256 public downSlippage = 100 * 1E14;

    /// @dev rebalanceUpCap is the maximum amount of xETH-stETH LP that can be burnt in a single rebalance
    uint256 public rebalanceUpCap;

    /// @dev rebalanceDownCap is the maximum amount of xETH that can be minted in a single rebalance
    uint256 public rebalanceDownCap;

    /// @dev lastRebalanceBlock is the block number of the last rebalance
    uint256 public lastRebalanceBlock;

    /// @dev cooldownBlocks is the number of blocks that must pass between rebalances
    uint256 public cooldownBlocks = 1800; /// (6 * 60 * 60) / 12

    /// @dev REBALANCE_UP_THRESHOLD is the upper threshold for the xETH-stETH LP ratio
    /// @notice if the ratio is above this value, rebalanceUp() will be called
    uint256 public REBALANCE_UP_THRESHOLD = 0.75E18;

    /// @dev REBALANCE_DOWN_THRESHOLD is the lower threshold for the xETH-stETH LP ratio
    /// @notice if the ratio is below this value, rebalanceDown() will be called
    uint256 public REBALANCE_DOWN_THRESHOLD = 0.68E18;

    /// @dev defender is the whitelisted bot that can call rebalance()
    address public defender;

    /// @dev cvxStaker is the CVX staking contract
    CVXStaker public cvxStaker;

    /// @dev afterCooldownPeriod is a modifier that checks if the cooldown period has passed
    modifier afterCooldownPeriod() {
        if (lastRebalanceBlock + cooldownBlocks >= block.number)
            revert CooldownNotFinished();
        _;
        lastRebalanceBlock = block.number;
    }

    constructor(
        address _xETH,
        address _stETH,
        address _curvePool,
        address _cvxStaker,
        uint256 _xETHIndex
    ) {
        if (
            _xETH == address(0) ||
            _stETH == address(0) ||
            _curvePool == address(0) ||
            _cvxStaker == address(0)
        ) {
            revert ZeroAddressProvided();
        }

        xETH = IXETH(_xETH);
        stETH = IERC20(_stETH);
        curvePool = ICurvePool(_curvePool);
        cvxStaker = CVXStaker(_cvxStaker);

        xETHIndex = _xETHIndex;
        stETHIndex = 1 - xETHIndex;

        _grantRole(DEFAULT_ADMIN_ROLE, msg.sender);
    }

    /// @dev preRebalanceCheck checks if a rebalance is even allowed based on pool ratios
    function preRebalanceCheck() internal view returns (bool isRebalanceUp) {
        uint256 stETHBal = curvePool.balances(stETHIndex);
        uint256 xETHBal = curvePool.balances(xETHIndex);

        /// @notice if either token balance is 0, the pool shall not be rebalanced
        if (stETHBal == 0 || xETHBal == 0) revert ZeroBalancePool();

        uint256 xEthPct = (xETHBal * BASE_UNIT) / (stETHBal + xETHBal);

        /// @notice if the ratio is above the upper threshold, rebalanceUp() will be called
        if (xEthPct > REBALANCE_UP_THRESHOLD) {
            isRebalanceUp = true;
        }
        /// @notice if the ratio is below the lower threshold, rebalanceDown() will be called
        /// @notice possible gas optimization here.
        else if (xEthPct < REBALANCE_DOWN_THRESHOLD) {
            isRebalanceUp = false;
        }
        /// @notice if the ratio is within the thresholds, the pool shall not be rebalanced
        else {
            revert RebalanceNotRequired();
        }
    }

    struct RebalanceUpQuote {
        uint256 lpBurn;
        uint256 min_xETHReceived;
    }

    /**
     * @dev Executes a rebalance up operation, which burns xETH-stETH LP to receive xETH.
     * @param quote The quote for the rebalance operation provided by the rebalance defender.
     * @return xETHReceived The amount of xETH received from the rebalance operation.
     * @notice Only the rebalance defender can call this function.
     * @notice The rebalance operation can only be performed after the cooldown period has elapsed.
     */
    function rebalanceUp(
        RebalanceUpQuote calldata quote
    )
        external
        onlyRole(REBALANCE_DEFENDER_ROLE)
        afterCooldownPeriod
        returns (uint256 xETHReceived)
    {
        if (quote.lpBurn == 0) revert ZeroValueProvided();

        bool isRebalanceUp = preRebalanceCheck();
        if (!isRebalanceUp) revert RebalanceUpNotAllowed();

        if (quote.lpBurn > rebalanceUpCap) revert RebalanceUpCapExceeded();

        uint256 min_xETHReceived = bestRebalanceUpQuote(quote);

        CVXStaker cachedCvxStaker = cvxStaker;

        uint256 amoLpBal = cachedCvxStaker.getTotalBalance();

        // if (amoLpBal == 0 || quote.lpBurn > amoLpBal) revert LpBalanceTooLow();
        if (quote.lpBurn > amoLpBal) revert LpBalanceTooLow();

        cachedCvxStaker.withdrawAndUnwrap(quote.lpBurn, false, address(this));

        xETHReceived = curvePool.remove_liquidity_one_coin(
            quote.lpBurn,
            int128(int(xETHIndex)),
            min_xETHReceived
        );

        xETH.burnShares(xETHReceived);

        emit RebalanceUpFinished(quote, xETHReceived);
    }

    struct RebalanceDownQuote {
        uint256 xETHAmount;
        uint256 minLpReceived;
    }

    /**
     * @dev Executes a rebalance down operation, which mints xETH and deposits into the Curve pool.
     * @param quote The quote for the rebalance operation provided by the rebalance defender.
     * @return lpAmountOut The amount of LP tokens received from the rebalance operation.
     * @notice Only the rebalance defender can call this function.
     * @notice The rebalance operation can only be performed after the cooldown period has elapsed.
     */
    function rebalanceDown(
        RebalanceDownQuote calldata quote
    )
        external
        onlyRole(REBALANCE_DEFENDER_ROLE)
        afterCooldownPeriod
        returns (uint256 lpAmountOut)
    {
        if (quote.xETHAmount == 0) revert ZeroValueProvided();

        bool isRebalanceUp = preRebalanceCheck();
        if (isRebalanceUp) revert RebalanceDownNotAllowed();

        if (quote.xETHAmount > rebalanceDownCap)
            revert RebalanceDownCapExceeded();

        uint256 minLpReceived = bestRebalanceDownQuote(quote);

        xETH.mintShares(quote.xETHAmount);

        uint256[2] memory amounts;
        amounts[xETHIndex] = quote.xETHAmount;

        IERC20(address(xETH)).approve(address(curvePool), quote.xETHAmount);

        lpAmountOut = curvePool.add_liquidity(amounts, minLpReceived);

        CVXStaker cachedCvxStaker = cvxStaker;

        IERC20(address(curvePool)).safeTransfer(
            address(cachedCvxStaker),
            lpAmountOut
        );
        cachedCvxStaker.depositAndStake(lpAmountOut);

        emit RebalanceDownFinished(quote, lpAmountOut);
    }

    /// @dev applySlippage applies the amount of slippage given 
    function applySlippage(uint256 amount, uint256 slippage) pure internal returns (uint256) {
        return slippage == 0 ? amount : (amount * (BASE_UNIT - slippage)) / BASE_UNIT;
    }

    /**
     * @dev Finds the best quote for rebalancing upwards.
     * @param defenderQuote The quote provided by the rebalance defender.
     * @return The best quote for rebalancing upwards.
     * @notice This function is internal and cannot be called outside of the contract.
     * @notice the defenderQuote should ideally be better than the contractQuote
     * @notice if its not, the contractQuote gets executed as a safeguard, reducing the risk of a large sandwich
     */
    function bestRebalanceUpQuote(
        RebalanceUpQuote calldata defenderQuote
    ) internal view returns (uint256) {
        // RebalanceUpQuote memory bestQuote;
        uint256 vp = curvePool.get_virtual_price();

        /// @dev first lets fill the bestQuote with the contractQuote
        // bestQuote.lpBurn = defenderQuote.lpBurn;
        uint256 min_xETHReceived = applySlippage(
            (vp * defenderQuote.lpBurn) / BASE_UNIT,
            upSlippage
        );

        if (defenderQuote.min_xETHReceived > min_xETHReceived)
            // bestQuote.min_xETHReceived = defenderQuote.min_xETHReceived 
            return defenderQuote.min_xETHReceived;

        return min_xETHReceived;
    }

    /**
     * @dev Finds the best quote for rebalancing downwards.
     * @param defenderQuote The quote provided by the rebalance defender.
     * @return The best quote for rebalancing downwards.
     * @notice the defenderQuote should ideally be better than the contractQuote
     * @notice if its not, the contractQuote gets executed as a safeguard, reducing the risk of a large sandwich
     */
    function bestRebalanceDownQuote(
        RebalanceDownQuote calldata defenderQuote
    ) internal view returns (uint256) {
        // RebalanceDownQuote memory bestQuote;
        uint256 vp = curvePool.get_virtual_price();

        /// @dev first lets fill the bestQuote with the contractQuote
        // bestQuote.xETHAmount = defenderQuote.xETHAmount;
        uint256 minLpReceived = applySlippage(
            (BASE_UNIT * defenderQuote.xETHAmount) / vp,
            downSlippage
        );

        if (defenderQuote.minLpReceived > minLpReceived)
            // bestQuote.minLpReceived = defenderQuote.minLpReceived;
            return defenderQuote.minLpReceived;

        return minLpReceived;
    }

    /**
     * @dev Sets the address of the rebalance defender.
     * @param newDefender The new rebalance defender address to be set.
     * @notice Only callable by a user with the DEFAULT_ADMIN_ROLE
     * @notice The new rebalance defender address cannot be set to the zero address.
     * @notice If a previous defender was set, their `REBALANCE_DEFENDER_ROLE` is revoked and transferred to the new defender.
     * @notice Emits a `DefenderUpdated` event.
     */
    function setRebalanceDefender(
        address newDefender
    ) external onlyRole(DEFAULT_ADMIN_ROLE) {
        if (newDefender == address(0)) revert ZeroAddressProvided();

        address cachedDefender = defender;

        if (cachedDefender != address(0)) {
            _revokeRole(REBALANCE_DEFENDER_ROLE, cachedDefender);
        }

        emit DefenderUpdated(cachedDefender, newDefender);

        defender = newDefender;
        _grantRole(REBALANCE_DEFENDER_ROLE, newDefender);
    }

    /**
     * @dev Sets the slippage in basis points for trading.
     * @param newUpSlippage The new maximum slippage in basis points for upward price movement to be set.
     * @param newDownSlippage The new maximum slippage in basis points for downward price movement to be set.
     * @notice 1 BPS = 1E14
     * @notice Only callable by a user with the DEFAULT_ADMIN_ROLE
     * @notice The new maximum slippage must be between 0.06% and 15% (in basis points).
     * @notice Emits a `SlippageUpdated` event.
     */
    function setSlippage(
      uint256 newUpSlippage,
      uint256 newDownSlippage
    ) external onlyRole(DEFAULT_ADMIN_ROLE) {
        if (newUpSlippage >= BASE_UNIT || newDownSlippage >= BASE_UNIT)
          revert InvalidSetSlippage();

        emit SlippageUpdated(upSlippage, newUpSlippage, downSlippage, newDownSlippage);

        upSlippage = newUpSlippage;
        downSlippage = newDownSlippage;
    }

    /**
     * @dev Sets the maximum burning cap (rebalanceUp) in a single transaction.
     * @param newRebalanceUpCap The new rebalance up cap to be set.
     * @notice Only callable by a user with the DEFAULT_ADMIN_ROLE
     * @notice The new rebalance up cap cannot be set to zero.
     * @notice Emits a `RebalanceUpCapUpdated` event.
     */
    function setRebalanceUpCap(
        uint256 newRebalanceUpCap
    ) external onlyRole(DEFAULT_ADMIN_ROLE) {
        if (newRebalanceUpCap == 0) revert ZeroValueProvided();

        emit RebalanceUpCapUpdated(rebalanceUpCap, newRebalanceUpCap);

        rebalanceUpCap = newRebalanceUpCap;
    }

    /**
     * @dev Sets the maximum minting cap (rebalanceDown) in a single transaction.
     * @param newRebalanceDownCap The new rebalance down cap to be set.
     * @notice Only callable by a user with the DEFAULT_ADMIN_ROLE
     * @notice The new rebalance down cap cannot be set to zero.
     * @notice Emits a `RebalanceDownCapUpdated` event.
     */
    function setRebalanceDownCap(
        uint256 newRebalanceDownCap
    ) external onlyRole(DEFAULT_ADMIN_ROLE) {
        if (newRebalanceDownCap == 0) revert ZeroValueProvided();

        emit RebalanceDownCapUpdated(rebalanceDownCap, newRebalanceDownCap);

        rebalanceDownCap = newRebalanceDownCap;
    }

    /**
     * @dev Sets the number of blocks for the unstake cooldown period
     * @param newCooldownBlocks The new number of blocks for the unstake cooldown period
     * @notice Only callable by a user with the DEFAULT_ADMIN_ROLE
     * @notice Emits a CooldownBlocksUpdated event with the old and new cooldown block values
     */
    function setCooldownBlocks(
        uint256 newCooldownBlocks
    ) external onlyRole(DEFAULT_ADMIN_ROLE) {
        if (newCooldownBlocks == 0) revert ZeroValueProvided();

        emit CooldownBlocksUpdated(cooldownBlocks, newCooldownBlocks);

        cooldownBlocks = newCooldownBlocks;
    }

    /**
     * @dev Sets the CVX staking contract address
     * @param _cvxStaker The address of the CVX staking contract
     * @notice Only callable by a user with the DEFAULT_ADMIN_ROLE
     * @notice The new CVX staker contract address cannot be set to the zero address.
     * @notice Emits a `CVXStakerUpdated` event.
     */
    function setCvxStaker(
        address _cvxStaker
    ) external onlyRole(DEFAULT_ADMIN_ROLE) {
        if (_cvxStaker == address(0)) revert ZeroAddressProvided();

        emit CVXStakerUpdated(address(cvxStaker), _cvxStaker);

        cvxStaker = CVXStaker(_cvxStaker);
    }

    /**
     * @dev Sets the threshold for triggering a `rebalanceUp` operation.
     * @param newRebalanceUpThreshold The new threshold to be set.
     * @notice Emits a `SetRebalanceUpThreshold` event with the old and new thresholds.
     * @notice Requires the caller to have the `DEFAULT_ADMIN_ROLE`.
     */
    function setRebalanceUpThreshold(
        uint256 newRebalanceUpThreshold
    ) external onlyRole(DEFAULT_ADMIN_ROLE) {
        emit SetRebalanceUpThreshold(
            REBALANCE_UP_THRESHOLD,
            newRebalanceUpThreshold
        );

        REBALANCE_UP_THRESHOLD = newRebalanceUpThreshold;
    }

    /**
     * @dev Sets the threshold for triggering a `rebalanceDown` operation.
     * @param newRebalanceDownThreshold The new threshold to be set.
     * @notice Emits a `SetRebalanceDownThreshold` event with the old and new thresholds.
     * @notice Requires the caller to have the `DEFAULT_ADMIN_ROLE`.
     */
    function setRebalanceDownThreshold(
        uint256 newRebalanceDownThreshold
    ) external onlyRole(DEFAULT_ADMIN_ROLE) {
        emit SetRebalanceDownThreshold(
            REBALANCE_DOWN_THRESHOLD,
            newRebalanceDownThreshold
        );

        REBALANCE_DOWN_THRESHOLD = newRebalanceDownThreshold;
    }

    /**
     * @dev Adds liquidity to the Curve pool using both xETH and stETH and stakes the resulting LP tokens in the CVX staking contract
     * @param stETHAmount The amount of stETH to be deposited
     * @param xETHAmount The amount of xETH to be deposited
     * @param minLpOut The minimum amount of LP tokens to receive from the Curve pool
     * @notice Transfers stETH and xETH from the caller to this contract, adds liquidity to the Curve pool, and stakes the resulting LP tokens in the CVX staking contract.
     * @notice Only callable by a user with the DEFAULT_ADMIN_ROLE
     * @return lpOut The amount of LP tokens received from the Curve pool
     */
    function addLiquidity(
        uint256 stETHAmount,
        uint256 xETHAmount,
        uint256 minLpOut
    ) external onlyRole(DEFAULT_ADMIN_ROLE) returns (uint256 lpOut) {
        stETH.safeTransferFrom(msg.sender, address(this), stETHAmount);
        xETH.mintShares(xETHAmount);

        uint256[2] memory amounts;

        amounts[xETHIndex] = xETHAmount;
        amounts[stETHIndex] = stETHAmount;

        IERC20(address(xETH)).approve(address(curvePool), xETHAmount);
        stETH.approve(address(curvePool), stETHAmount);

        lpOut = curvePool.add_liquidity(amounts, minLpOut);

        /// @notice no need for safeApprove, direct transfer + deposit
        IERC20(address(curvePool)).safeTransfer(address(cvxStaker), lpOut);
        cvxStaker.depositAndStake(lpOut);
    }

    /**
     * @notice Adds liquidity only with stETH and stakes the resulting LP tokens in the cvxCRV staking contract.
     * @param stETHAmount The amount of stETH to add as liquidity.
     * @param minLpOut The minimum expected amount of LP tokens to receive.
     * @return lpOut The actual amount of LP tokens received.
     */
    function addLiquidityOnlyStETH(
        uint256 stETHAmount,
        uint256 minLpOut
    ) external onlyRole(DEFAULT_ADMIN_ROLE) returns (uint256 lpOut) {
        stETH.safeTransferFrom(msg.sender, address(this), stETHAmount);

        uint256[2] memory amounts;

        amounts[stETHIndex] = stETHAmount;

        stETH.approve(address(curvePool), stETHAmount);

        lpOut = curvePool.add_liquidity(amounts, minLpOut);

        /// @notice no need for safeApprove, direct transfer + deposit
        IERC20(address(curvePool)).safeTransfer(address(cvxStaker), lpOut);
        cvxStaker.depositAndStake(lpOut);
    }

    /**
     * @dev Removes liquidity from the Curve pool using both xETH and stETH and transfers the resulting tokens to the caller
     * @param lpAmount The amount of LP tokens to be burned
     * @param minStETHOut The minimum amount of stETH to receive from the Curve pool
     * @param minXETHOut The minimum amount of xETH to receive from the Curve pool
     * @notice Checks if the AMO owns enough LP tokens, withdraws and unwraps them, and removes liquidity from the Curve pool.
     *      The resulting xETH and stETH are then transferred to the caller.
     * @notice Only callable by a user with the DEFAULT_ADMIN_ROLE
     * @return outputs An array containing the resulting amounts of xETH and stETH received from the Curve pool
     */
    function removeLiquidity(
        uint256 lpAmount,
        uint256 minStETHOut,
        uint256 minXETHOut
    )
        external
        onlyRole(DEFAULT_ADMIN_ROLE)
        returns (uint256[2] memory outputs)
    {
        /// @dev check if AMO owns enough LP
        uint256 amoBalance = cvxStaker.getTotalBalance();

        if (lpAmount > amoBalance) {
            revert LpBalanceTooLow();
        }

        cvxStaker.withdrawAndUnwrap(lpAmount, false, address(this));

        uint256[2] memory minAmounts;

        minAmounts[xETHIndex] = minXETHOut;
        minAmounts[stETHIndex] = minStETHOut;

        outputs = curvePool.remove_liquidity(lpAmount, minAmounts);

        xETH.burnShares(outputs[xETHIndex]);
        stETH.safeTransfer(msg.sender, outputs[stETHIndex]);
    }

    /**
     * @dev Removes liquidity from the Curve pool using only stETH and transfers the resulting stETH to the caller
     * @param lpAmount The amount of LP tokens to be burned
     * @param minStETHOut The minimum amount of stETH to receive from the Curve pool
     * @notice Checks if the AMO owns enough LP tokens, withdraws and unwraps them, and removes liquidity from the Curve pool.
     *      The resulting stETH is then transferred to the caller.
     * @notice Only callable by a user with the DEFAULT_ADMIN_ROLE
     */
    function removeLiquidityOnlyStETH(
        uint256 lpAmount,
        uint256 minStETHOut
    ) external onlyRole(DEFAULT_ADMIN_ROLE) {
        /// @dev check if AMO owns enough LP
        uint256 amoBalance = cvxStaker.getTotalBalance();

        if (lpAmount > amoBalance) {
            revert LpBalanceTooLow();
        }

        cvxStaker.withdrawAndUnwrap(lpAmount, false, address(this));

        uint256[2] memory minAmounts;

        minAmounts[stETHIndex] = minStETHOut;

        uint256 output = curvePool.remove_liquidity_one_coin(
            lpAmount,
            int128(int(stETHIndex)),
            minStETHOut
        );

        stETH.safeTransfer(msg.sender, output);
    }

    /**
     * @notice Recover any token from AMO
     * @param token Token to recover
     * @param to Recipient address
     * @param amount Amount to recover
     */
    function recoverToken(
        address token,
        address to,
        uint256 amount
    ) external onlyRole(DEFAULT_ADMIN_ROLE) {
        IERC20(token).safeTransfer(to, amount);

        emit RecoveredToken(token, to, amount);
    }
}